# $Id: PKGBUILD 182608 2013-04-10 23:44:08Z bisson $
# Maintainer: Gaetan Bisson <bisson@archlinux.org>
# Contributor: Angel Velasquez <angvp@archlinux.org>
# Contributor: Andrea Scarpino <andrea@archlinux.org>
# Contributor: Alexander Fehr <pizzapunk gmail com>
# Contributor: Link Dupont <link@subpop.net>

pkgname=mpc
pkgver=0.23
pkgrel=1
pkgdesc='Minimalist command line interface to MPD'
url='http://www.musicpd.org/clients/mpc/'
license=('GPL2')
arch=('i686' 'x86_64')
depends=('libmpdclient')
source=("http://www.musicpd.org/download/${pkgname}/${pkgver%.*}/${pkgname}-${pkgver}.tar.bz2")
sha1sums=('25d217b699fd70fd9b65b8b820addf794a109a9c')

options=('!emptydirs')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	./configure --prefix=/usr 
	make 
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install 
	install -Dm644 doc/mpc-completion.bash "${pkgdir}/usr/share/bash-completion/completions/mpc" 
}
